

Définition de syntaxes concrètes textuelles

La version d'Eclipse à utiliser est la suivante :

```
/mnt/n7fs/ens/tp_babin/gls/eclipse-gls/bin/eclipse-gls
```

Nous avons vu comment définir la syntaxe abstraite d'un DSML. Le principe est d'utiliser Ecore (ou un autre langage de métamodélisation) pour décrire le métamodèle et OCL (ou un autre langage de contrainte) pour exprimer les propriétés liées à la sémantique statique qui n'ont pas pu être capturées par le métamodèle.

La syntaxe abstraite ne peut pas être manipulée directement. Le projet EMF d'Eclipse permet d'engendrer un éditeur arborescent et les classes Java qui permettront de manipuler un modèle conforme à un métamodèle. Cependant, ce ne sont pas des outils très pratiques lorsque l'on veut saisir un modèle. Aussi, il est souhaitable d'associer à une syntaxe abstraite une ou plusieurs syntaxes concrètes pour faciliter la construction et la modification des modèles. Ces syntaxes concrètes peuvent être textuelles (l'objet de cette séance) ou graphiques (séance précédente).

Pour la définition des syntaxes concrètes textuelles, nous allons utiliser l'outil *Xtext*¹ de openArchitectureWare (oAW). *Xtext* fait partie du projet TMF (Textual Modeling Framework). Il permet non seulement de définir une syntaxe textuelle pour un DSL mais aussi de disposer, au travers d'Eclipse, d'un environnement de développement pour ce langage, avec en particulier un éditeur syntaxique (coloration, complétion, outline, détection et visualisation des erreurs, etc).

Xtext s'appuie sur un générateur d'analyseurs descendants récursifs (LL(k)). L'idée est de décrire à la fois la syntaxe concrète du langage et comment le modèle EMF est construit en mémoire au fur et à mesure de l'analyse syntaxique.

1 Définir des syntaxes concrètes textuelles pour SimplePDL

Xtext consiste à définir une syntaxe concrète textuelle pour un DSML au travers d'une grammaire qui doit pouvoir être analysée en LL(k). En particulier, elle ne doit pas être récursive à gauche² ! Généralement, *Xtext* est utilisé pour engendrer le métamodèle correspondant. C'est ce que nous allons voir dans ce sujet. *Xtext* peut aussi travailler à partir d'un métamodèle existant. Cependant, il ne faut pas que la distance soit trop grande entre le métamodèle existant et la syntaxe textuelle souhaitée sinon il est conseillé de définir un métamodèle adapté à la syntaxe concrète (par exemple engendré automatiquement) puis d'écrire une transformation de modèle pour obtenir un modèle conforme au métamodèle existant du DSML.

1. <https://www.eclipse.org/Xtext/>

2. Voir la documentation de *Xtext* : <http://www.eclipse.org/Xtext/documentation>

Exercice 1 : Première syntaxe textuelle pour SimplePDL : PDL1

Comme toujours avec Eclipse, il faut commencer par créer un nouveau projet.

1.1 Création du projet Xtext. Sous Eclipse, on commence par définir un projet (New > Project, puis Xtext > Xtext project). Il faut indiquer d'abord le nom du projet principal (fr.enseeiht.pdl1 par exemple³), le nom du langage (fr.enseeiht.PDL1), et l'extension pour les fichiers contenant les modèles correspondants (pd11).

Plusieurs projets sont alors créés, en particulier fr.enseeiht.pdl1 qui contient dans src le fichier *Xtext* qui décrit la syntaxe de PDL1. Ce fichier nommé pd11.xtext apparaît d'ailleurs dans l'éditeur. Il contient un squelette montrant la structure d'un fichier *Xtext*. La première ligne indique le langage qui est défini. La ligne suivante indique le métamodèle qui sera engendré. Le reste du fichier est composé des règles de production définissant la syntaxe concrète textuelle avec les actions qui permettent de construire le modèle EMF correspondant.

1.2 Définition de la syntaxe PDL1. Un exemple de la syntaxe textuelle souhaitée est donné dans l'exemple ci-dessous.

```
process dvp {  
    wd a  
    wd b  
    wd c  
    ws s2s from a to b  
    ws f2f from b to c  
}
```

Le fichier *Xtext* correspondant est donné au listing 1.

Remplacer le texte de PDL1.xtext par celui de la figure 1.

1.3 Engendrer l'outillage Xtext pour PDL1. Dans le répertoire src du projet principal, il faut sélectionner le fichier GeneratePDL1.mwe2, cliquer à droite et sélectionner *Run As > MWE Workflow*. La génération prend un peu de temps...

1.4 Utiliser l'éditeur PDL1 engendré. Il faut déployer les greffons *Xtext* engendrés comment cela a été fait dans les séances précédentes. Ensuite, vous pouvez créer dans le projet contenant les exemples fr.enseeiht.simplepdl.exemples un fichier avec l'extension .pd11, par exemple process1.pd11. Saisir un exemple de processus et regarder les possibilités de l'éditeur (colorisation, complétion syntaxique...).

1.5 Consulter le métamodèle PDL1 engendré. Le métamodèle PDL1 a été engendré. Le fichier PDL1.ecore est présent dans le répertoire model/generated du projet principal. Pour l'ouvrir avec l'éditeur graphique, cliquer à droite et sélectionner *Initialize Ecore Diagram* Ensuite cliquer sur *Finish*. Puis sélectionner le viewpoint *Design/Entities*. Enfin choisir le package *pDL1* du fichier *PDL1.ecore*.

Exercice 2 : Autre syntaxe textuelle pour SimplePDL : PDL2

On considère une nouvelle syntaxe textuelle dont le fichier *Xtext* est donné au listing 2.

2.1 Donner la syntaxe PDL2 du processus donné à la question 1.2.

3. Le nom du projet doit être en minuscules et contenir au moins un point « . ».

Listing 1 – Description *Xtext* pour la syntaxe PDL1

```
1 grammar fr.enseeiht.PDL1 with org.eclipse.xtext.common.Terminals
2
3 generate pDL1 "http://www.enseeiht.fr/PDL1"
4
5 Process :
6     'process' name=ID '{'
7         processElements+=ProcessElement*
8     '}'
9     ;
10
11 ProcessElement :
12     WorkDefinition
13     | WorkSequence
14     | Guidance
15     ;
16
17 WorkSequence :
18     'ws' linkType=WorkSequenceType
19         'from' predecessor=[WorkDefinition]
20         'to' successor=[WorkDefinition]
21     ;
22
23 Guidance:
24     text=STRING
25     ;
26
27 WorkDefinition :
28     'wd' name=ID
29     ;
30
31 enum WorkSequenceType :
32     start2start = 's2s'
33     | finish2start = 'f2s'
34     | start2finish = 's2f'
35     | finish2finish = 'f2f'
36     ;
```

2.2 Engendrer l'éditeur pour PDL2, le déployer et l'utiliser.

2.3 Comparer les deux métamodèles, PDL1.ecore et PDL2.ecore.

Exercice 3 : Engendrer la description textuelle à partir du modèle

Voyons comment utiliser un métamodèle existant plutôt que de laisser *Xtext* l'engendrer.

3.1 S'assurer de l'existence du fichier SimplePDL.genmodel associé au méta-modèle SimplePDL.ecore.

Puis, enregistrer le méta-modèle *SimplePDL* dans le registre d'Eclipse.

3.2 Créer un projet *Xtext Project From Existing Ecore Models*. Ajouter (bouton *Add*) SimplePDL.genmodel aux EPackages, et choisir Process comme *Entry rule*, puis cliquer sur *Next*. Nommer le projet fr.enseeiht.simplepdl.txt, avec fr.enseeiht.simplepdl.SimplePDLDSL comme nom de langage et pdlt comme extension. Décocher *Create SDK feature project*, puis cliquer sur *Finish*.

Modifier les règles de la grammaire comme sur le listing 3.

Il suffit alors d'exécuter le fichier GenerateSimplePDLdsl.mwe2, puis de déployer les greffons engendrés pour utiliser l'éditeur engendré.

2 Application aux réseaux de Petri

Exercice 4 : Définir des syntaxes textuelles pour les réseaux de Petri

Proposer une ou plusieurs syntaxes textuelles pour les réseaux de Petri et utiliser *Xtext* pour engendrer l'outillage Eclipse correspondant.

Listing 2 – Description *Xtext* pour la syntaxe PDL2

```
1 grammar fr.enseeiht.PDL2 with org.eclipse.xtext.common.Terminals
2
3 generate pDL2 "http://www.enseeiht.fr/PDL2"
4
5 Process :
6     'process' name=ID '{'
7         processElements+=ProcessElement*
8     '}'
9     ;
10
11 ProcessElement :
12     WorkDefinition
13     | Guidance
14     ;
15
16 WorkDefinition :
17     'wd' name=ID '{'
18         ('starts' 'if' (linksToPredecessors+=DependanceStart)+)?
19         ('finishes' 'if' (linksToPredecessors+=DependanceFinish)+)?
20     '}'
21     ;
22
23 DependanceStart :
24     predecessor=[WorkDefinition] link=WorkSequenceKindStart
25     ;
26
27 WorkSequenceKindStart:
28     Started2Start = 'started'
29     | Started2Finish = 'finished'
30     ;
31
32 DependanceFinish :
33     predecessor=[WorkDefinition] link=WorkSequenceKindFinish
34     ;
35
36 WorkSequenceKindFinish:
37     Finished2Start = 'started'
38     | Finished2Finish = 'finished'
39     ;
40
41 Guidance:
42     text=STRING
43     ;
```

Listing 3 – Syntaxe textuelle *Xtext* utilisant le métamodèle SimplePDL existant

```
grammar fr.enseeiht.txt.SimplePDL with org.eclipse.xtext.common.Terminals
import "platform:/resource/fr.enseeiht.simplepdl/SimplePDL.ecore" as pdl
// generate simplePDL "http://www.enseeiht.fr/txt/SimplePDL"
```

```
Process returns pdl::Process :
    'process' name=ID '{'
    processElements+=ProcessElement*
    '}'
    ;
```

```
ProcessElement returns pdl::ProcessElement :
    WorkDefinition
    | WorkSequence
    | Guidance
    ;
```

```
WorkDefinition returns pdl::WorkDefinition :
    'wd' name=ID
    ;
```

```
WorkSequence returns pdl::WorkSequence :
    'ws' linkType=WorkSequenceType
        'from' predecessor=[pdl::WorkDefinition]
        'to' successor=[pdl::WorkDefinition]
    ;
```

```
Guidance returns pdl::Guidance :
    text=STRING
    ;
```

```
enum WorkSequenceType returns pdl::WorkSequenceType :
    startToStart = 's2s'
    | finishToStart = 'f2s'
    | startToFinish = 's2f'
    | finishToFinish = 'f2f'
    ;
```